Pensieve header: Implementing ordering symbols for U(g0).

Reminders



$R = \sum a_i \otimes b_i \in A \otimes A = \hat{U}(\mathfrak{g}) \otimes \hat{U}(\mathfrak{g})$

s.t. $R^{12}R^{13}R^{23} = R^{23}R^{13}R^{12}$

$\to \sum_{i,j,k} b_i a_j b_k a_i b_j a_k \in U(\mathfrak{g})$



PBW: $\mathfrak{g} = \langle x_1 \dots x_k \rangle \Rightarrow \{x_1^{a_1} x_2^{a_2} \dots x_k^{a_k} : a_i \in \mathbb{Z}_{\geq 0}\}$ is a basis of $U(\mathfrak{g})$.

Today: $\mathfrak{g}_0 = \langle h, e, l, f \rangle \big/ \begin{smallmatrix} h \text{ central} \\ [e,l] = -e \quad [f,l] = f \quad [e,f] = h \end{smallmatrix}$

$r = h \otimes l + e \otimes f \qquad R = \exp(r)$

Note $U(\mathfrak{g}_0)^{\otimes S} = U(\bigoplus_S \mathfrak{g}_0) = U\left(\langle h_i, e_i, l_i, f_i \rangle \big/ \begin{smallmatrix} h_i \text{ central} \\ [e_i, l_j] = \delta_{ij} e_i \text{ etc.} \end{smallmatrix}\right)$

Implementing g_0

```
PBWRule = {e → 1, l → 2, f → 3};
B[U@e, U@l] = -U@e;
B[U@f, U@l] = U@f;
B[U@e, U@f] = h U[];
```

```
$TD = 3;
ħ /: ħ^d_. /; d > $TD := 0;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule];
x_ ≺ y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {h → h_i, t → t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] := B[U[y], U[x]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] := If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[];
UU[l_, r___] := U[l] ** UU[r];
```

```
UProducts[{}, 0] = {UU[]};
UProducts[{}, n_Integer] /; n > 0 = {};
UProducts[{x_, xs___}, n_Integer] :=
  Sort@Flatten@Table[UU[x^k] ** u, {k, 0, n}, {u, UProducts[{xs}, n - k]}];
UProducts[xs_List, k_Integer, n_Integer] := UProducts[Flatten@Table[x_j, {x, xs}, {j, k}], n];
UProducts[any__, {n_}] := Flatten@Table[UProducts[any, k], {k, 0, n}];
```

```
r_{i_,j_} := Simp[ℏ (h_i UU[l_j] + UU[e_i, f_j])]
```

```
UExp[u_] := Module[{s, t, k},
   s = t = U[]; k = 0;
   While[k < 20 ∧ 0 =!= (t = t ** u), s += t / (++k) !];
   Simp[s]
  ];
R_{i_,j_} := UExp[r_{i,j}];
```

```
$TD = 3; Simp[R_{1,2} ** R_{1,3} ** R_{2,3} - R_{2,3} ** R_{1,3} ** R_{1,2}]
```

0

old above / new below

The “Internal Multiplication” and Meta-Associativity

```
m[i_, j_, k_][ℰ_] := Simp[ℰ /. {
    u_U :> UU @@ Join[DeleteCases[u, x_{i|j}], U @@ Cases[u, x_i :> x_k], U @@ Cases[u, x_j :> x_k]],
    h_{i|j} → h_k
   }]
```

```
UU[e_1, l_4, f_2]
```

$U[e_1, l_4, f_2]$

```
UU[e_1, l_4, f_2] // m[1, 2, 3]
```

$U[e_3, l_4, f_3]$

```
UU[e_1, l_4, f_2] // m[2, 1, 3]
```

$-h_3 U[l_4] + U[e_3, l_4, f_3]$

```
Union@Table[
  (u // m[1, 2, 1] // m[1, 3, 1]) - (u // m[2, 3, 2] // m[1, 2, 1]),
  {u, UProducts[{e, l, f}, 4, {3}]}
 ]
```

{0}

The Invariant of the Trefoil




```
$TD = 2; R4,1 ** R2,5 ** R6,3
```

$$U[\,] + \hbar h_4 U[l_1] + \hbar h_6 U[l_3] + \hbar h_2 U[l_5] + \left(\hbar + \frac{\hbar^2 h_2}{2}\right) U[e_2, f_5] + \left(\hbar + \frac{\hbar^2 h_4}{2}\right) U[e_4, f_1] + \left(\hbar + \frac{\hbar^2 h_6}{2}\right) U[e_6, f_3] +$$
$$\frac{1}{2}\hbar^2 h_4^2 U[l_1, l_1] + \hbar^2 h_4 h_6 U[l_1, l_3] + \hbar^2 h_2 h_4 U[l_1, l_5] + \frac{1}{2}\hbar^2 h_6^2 U[l_3, l_3] + \hbar^2 h_2 h_6 U[l_3, l_5] + \frac{1}{2}\hbar^2 h_2^2 U[l_5, l_5] +$$
$$\hbar^2 h_4 U[e_2, l_1, f_5] + \hbar^2 h_6 U[e_2, l_3, f_5] + \hbar^2 h_2 U[e_2, l_5, f_5] + \hbar^2 h_4 U[e_4, l_1, f_1] + \hbar^2 h_6 U[e_4, l_3, f_1] +$$
$$\hbar^2 h_2 U[e_4, l_5, f_1] + \hbar^2 h_4 U[e_6, l_1, f_3] + \hbar^2 h_6 U[e_6, l_3, f_3] + \hbar^2 h_2 U[e_6, l_5, f_3] + \frac{1}{2}\hbar^2 U[e_2, e_2, f_5, f_5] +$$
$$\hbar^2 U[e_2, e_4, f_1, f_5] + \hbar^2 U[e_2, e_6, f_3, f_5] + \frac{1}{2}\hbar^2 U[e_4, e_4, f_1, f_1] + \hbar^2 U[e_4, e_6, f_1, f_3] + \frac{1}{2}\hbar^2 U[e_6, e_6, f_3, f_3]$$

```
$TD = 2; R4,1 ** R2,5 ** R6,3 // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // m[1, 5, 1] // m[1, 6, 1]
```

$$\left(1 - 2\hbar h_1 + \hbar^2 h_1^2\right) U[\,] + \left(3\hbar h_1 - 6\hbar^2 h_1^2\right) U[l_1] +$$
$$\left(3\hbar - \frac{3\hbar^2 h_1}{2}\right) U[e_1, f_1] + \frac{9}{2}\hbar^2 h_1^2 U[l_1, l_1] + 9\hbar^2 h_1 U[e_1, l_1, f_1] + \frac{9}{2}\hbar^2 U[e_1, e_1, f_1, f_1]$$

```
Timing[$TD = 3; R4,1 ** R2,5 ** R6,3 // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // m[1, 5, 1] // m[1, 6, 1]]
```

$$\Big\{0.59375,\ \left(1 - 2\hbar h_1 + \hbar^2 h_1^2 + \frac{2}{3}\hbar^3 h_1^3\right) U[\,] +$$
$$\left(3\hbar h_1 - 6\hbar^2 h_1^2 + 3\hbar^3 h_1^3\right) U[l_1] + \left(3\hbar - \frac{3\hbar^2 h_1}{2} - \frac{3}{2}\hbar^3 h_1^2\right) U[e_1, f_1] + \left(\frac{9}{2}\hbar^2 h_1^2 - 9\hbar^3 h_1^3\right) U[l_1, l_1] +$$
$$\left(9\hbar^2 h_1 - \frac{9}{2}\hbar^3 h_1^2\right) U[e_1, l_1, f_1] + \frac{9}{2}\hbar^3 h_1^3 U[l_1, l_1, l_1] + \left(\frac{9\hbar^2}{2} + \frac{9\hbar^3 h_1}{2}\right) U[e_1, e_1, f_1, f_1] +$$
$$\frac{27}{2}\hbar^3 h_1^2 U[e_1, l_1, l_1, f_1] + \frac{27}{2}\hbar^3 h_1 U[e_1, e_1, l_1, f_1, f_1] + \frac{9}{2}\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]\Big\}$$

Ordering Symbols

```
𝕆[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ℏ, 0, $TD}], vs] /. (p_ → c_) ⧴ c UU @@ (us^p)]
 ]
```

Theorem. $R = \mathbb{O}\left(\exp\left(hl + \frac{e^h - 1}{h} ef\right) \mid e \otimes lf\right)$.

Brute Proof.

```
$TD = 6; 𝕆[Exp[ℏ h1 l2 + (e^(ℏ h1) - 1)/h1 e1 f2], {e1} → 1, {l2, f2} → 2] == R1,2
```

```
True
```

$$\text{Timing}\Big[\$TD = 3;\ \mathbb{O}\Big[\text{Exp}\Big[\hbar\, h\, l_1 + \frac{e^{\hbar h}-1}{h} e_4 f_1 + \hbar\, h\, l_5 + \frac{e^{\hbar h}-1}{h} e_2 f_5 + \hbar\, h\, l_3 + \frac{e^{\hbar h}-1}{h} e_6 f_3\Big],\ \{l_1, f_1, e_2, l_3, f_3, e_4, l_5, f_5, e_6\} \to 1\Big]\ /.\ h_1 \to h\Big]$$

$$\Big\{0.0625,\ \Big(1 - 2 h\hbar + h^2\hbar^2 + \frac{2h^3\hbar^3}{3}\Big) U[\,] +$$
$$\big(3h\hbar - 6h^2\hbar^2 + 3h^3\hbar^3\big) U[l_1] + \Big(3\hbar - \frac{3h\hbar^2}{2} - \frac{3h^2\hbar^3}{2}\Big) U[e_1, f_1] + \Big(\frac{9h^2\hbar^2}{2} - 9h^3\hbar^3\Big) U[l_1, l_1] +$$
$$\Big(9h\hbar^2 - \frac{9h^2\hbar^3}{2}\Big) U[e_1, l_1, f_1] + \frac{9}{2} h^3\hbar^3 U[l_1, l_1, l_1] + \Big(\frac{9\hbar^2}{2} + \frac{9h\hbar^3}{2}\Big) U[e_1, e_1, f_1, f_1] +$$
$$\frac{27}{2} h^2\hbar^3 U[e_1, l_1, l_1, f_1] + \frac{27}{2} h\hbar^3 U[e_1, e_1, l_1, f_1, f_1] + \frac{9}{2}\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]\Big\}$$

$$\text{Timing}\Big[\$TD = 6;\ \mathbb{O}\Big[\text{Exp}\Big[\hbar\, h\, l_1 + \frac{e^{\hbar h}-1}{h} e_4 f_1 + \hbar\, h\, l_5 + \frac{e^{\hbar h}-1}{h} e_2 f_5 + \hbar\, h\, l_3 + \frac{e^{\hbar h}-1}{h} e_6 f_3\Big],\ \{l_1, f_1, e_2, l_3, f_3, e_4, l_5, f_5, e_6\} \to 1\Big]\ /.\ h_1 \to h\Big]$$

$$\Big\{3.42188,\ \Big(1 - 2h\hbar + h^2\hbar^2 + \frac{2h^3\hbar^3}{3} - \frac{5h^4\hbar^4}{12} - \frac{23h^5\hbar^5}{30} + \frac{151h^6\hbar^6}{360}\Big) U[\,] +$$
$$\Big(3h\hbar - 6h^2\hbar^2 + 3h^3\hbar^3 + 2h^4\hbar^4 - \frac{5h^5\hbar^5}{4} - \frac{23h^6\hbar^6}{10}\Big) U[l_1] +$$
$$\Big(3\hbar - \frac{3h\hbar^2}{2} - \frac{3h^2\hbar^3}{2} + \frac{7h^3\hbar^4}{8} + \frac{61h^4\hbar^5}{40} - \frac{67h^5\hbar^6}{80}\Big) U[e_1, f_1] +$$
$$\Big(\frac{9h^2\hbar^2}{2} - 9h^3\hbar^3 + \frac{9h^4\hbar^4}{2} + 3h^5\hbar^5 - \frac{15h^6\hbar^6}{8}\Big) U[l_1, l_1] +$$
$$\Big(9h\hbar^2 - \frac{9h^2\hbar^3}{2} - \frac{9h^3\hbar^4}{2} + \frac{21h^4\hbar^5}{8} + \frac{183h^5\hbar^6}{40}\Big) U[e_1, l_1, f_1] + \Big(\frac{9h^3\hbar^3}{2} - 9h^4\hbar^4 + \frac{9h^5\hbar^5}{2} + 3h^6\hbar^6\Big) U[l_1, l_1, l_1] +$$
$$\Big(\frac{9\hbar^2}{2} + \frac{9h\hbar^3}{2} + \frac{9h^2\hbar^4}{8} - \frac{3h^3\hbar^5}{8} + \frac{111h^4\hbar^6}{80}\Big) U[e_1, e_1, f_1, f_1] +$$
$$\Big(\frac{27h^2\hbar^3}{2} - \frac{27h^3\hbar^4}{4} - \frac{27h^4\hbar^5}{4} + \frac{63h^5\hbar^6}{16}\Big) U[e_1, l_1, l_1, f_1] + \Big(\frac{27h^4\hbar^4}{8} - \frac{27h^5\hbar^5}{4} + \frac{27h^6\hbar^6}{8}\Big) U[l_1, l_1, l_1, l_1] +$$
$$\Big(\frac{27h\hbar^3}{2} + \frac{27h^2\hbar^4}{2} + \frac{27h^3\hbar^5}{8} - \frac{9h^4\hbar^6}{8}\Big) U[e_1, e_1, l_1, f_1, f_1] + \Big(\frac{27h^3\hbar^4}{2} - \frac{27h^4\hbar^5}{4} - \frac{27h^5\hbar^6}{4}\Big) U[e_1, l_1, l_1, l_1, f_1] +$$
$$\Big(\frac{81h^5\hbar^5}{40} - \frac{81h^6\hbar^6}{20}\Big) U[l_1, l_1, l_1, l_1, l_1] + \Big(\frac{9\hbar^3}{2} + \frac{45h\hbar^4}{4} + \frac{117h^2\hbar^5}{8} + \frac{105h^3\hbar^6}{8}\Big) U[e_1, e_1, e_1, f_1, f_1, f_1] +$$
$$\Big(\frac{81h^2\hbar^4}{4} + \frac{81h^3\hbar^5}{4} + \frac{81h^4\hbar^6}{16}\Big) U[e_1, e_1, l_1, l_1, f_1, f_1] + \Big(\frac{81h^4\hbar^5}{8} - \frac{81h^5\hbar^6}{16}\Big) U[e_1, l_1, l_1, l_1, l_1, f_1] +$$
$$\frac{81}{80} h^6\hbar^6 U[l_1, l_1, l_1, l_1, l_1, l_1] + \Big(\frac{27h\hbar^4}{2} + \frac{135h^2\hbar^5}{4} + \frac{351h^3\hbar^6}{8}\Big) U[e_1, e_1, e_1, l_1, f_1, f_1, f_1] +$$
$$\Big(\frac{81h^3\hbar^5}{4} + \frac{81h^4\hbar^6}{4}\Big) U[e_1, e_1, l_1, l_1, l_1, f_1, f_1] + \frac{243}{40} h^5\hbar^6 U[e_1, l_1, l_1, l_1, l_1, l_1, f_1] +$$
$$\Big(\frac{27\hbar^4}{8} + \frac{27h\hbar^5}{2} + \frac{459h^2\hbar^6}{16}\Big) U[e_1, e_1, e_1, e_1, f_1, f_1, f_1, f_1] +$$
$$\Big(\frac{81h^2\hbar^5}{4} + \frac{405h^3\hbar^6}{8}\Big) U[e_1, e_1, e_1, l_1, l_1, f_1, f_1, f_1] + \frac{243}{16} h^4\hbar^6 U[e_1, e_1, l_1, l_1, l_1, l_1, f_1, f_1] +$$
$$\Big(\frac{81h\hbar^5}{8} + \frac{81h^2\hbar^6}{2}\Big) U[e_1, e_1, e_1, e_1, l_1, f_1, f_1, f_1, f_1] + \frac{81}{4} h^3\hbar^6 U[e_1, e_1, e_1, l_1, l_1, l_1, f_1, f_1, f_1] +$$
$$\Big(\frac{81\hbar^5}{40} + \frac{891h\hbar^6}{80}\Big) U[e_1, e_1, e_1, e_1, e_1, f_1, f_1, f_1, f_1, f_1] + \frac{243}{16} h^2\hbar^6 U[e_1, e_1, e_1, e_1, l_1, l_1, f_1, f_1, f_1, f_1] +$$
$$\frac{243}{40} h\hbar^6 U[e_1, e_1, e_1, e_1, e_1, l_1, f_1, f_1, f_1, f_1, f_1] + \frac{81}{80}\hbar^6 U[e_1, e_1, e_1, e_1, e_1, e_1, f_1, f_1, f_1, f_1, f_1, f_1]\Big\}$$

The Big g_0 Lemma.

1. $\mathbb{O}\big(e^{\gamma l+\beta e} \mid le\big) = \mathbb{O}\big(e^{\gamma l+e^{\gamma}\beta e} \mid el\big)$.
2. $\mathbb{O}\big(e^{\gamma l+\beta f} \mid fe\big) = \mathbb{O}\big(e^{\gamma l+e^{\gamma}\beta f} \mid lf\big)$.
3. $\mathbb{O}\big(e^{\beta e+\alpha f+\delta ef} \mid fe\big) = \mathbb{O}\big(\nu e^{\nu(-\alpha\beta h+\beta e+\alpha f+\delta ef)} \mid ef\big)$.

Brute Proof.

$TD = 3; $\mathbb{O}\left[e^{\hbar\gamma l_1+\beta\hbar e_1}, \{l_1, e_1\}\to 1\right]$

$U[\,] + \left(\beta\hbar + \beta\gamma\hbar^2 + \frac{1}{2}\beta\gamma^2\hbar^3\right)U[e_1] + \gamma\hbar U[l_1] + \left(\frac{\beta^2\hbar^2}{2} + \beta^2\gamma\hbar^3\right)U[e_1, e_1] + \left(\beta\gamma\hbar^2 + \beta\gamma^2\hbar^3\right)U[e_1, l_1] + \frac{1}{2}\gamma^2\hbar^2 U[l_1, l_1] + \frac{1}{6}\beta^3\hbar^3 U[e_1, e_1, e_1] + \frac{1}{2}\beta^2\gamma\hbar^3 U[e_1, e_1, l_1] + \frac{1}{2}\beta\gamma^2\hbar^3 U[e_1, l_1, l_1] + \frac{1}{6}\gamma^3\hbar^3 U[l_1, l_1, l_1]$

$TD = 3; $\mathbb{O}\left[e^{\hbar\gamma l_1+e^{\hbar\gamma}\beta\hbar e_1}, \{e_1, l_1\}\to 1\right]$

$U[\,] + \left(\beta\hbar + \beta\gamma\hbar^2 + \frac{1}{2}\beta\gamma^2\hbar^3\right)U[e_1] + \gamma\hbar U[l_1] + \left(\frac{\beta^2\hbar^2}{2} + \beta^2\gamma\hbar^3\right)U[e_1, e_1] + \left(\beta\gamma\hbar^2 + \beta\gamma^2\hbar^3\right)U[e_1, l_1] + \frac{1}{2}\gamma^2\hbar^2 U[l_1, l_1] + \frac{1}{6}\beta^3\hbar^3 U[e_1, e_1, e_1] + \frac{1}{2}\beta^2\gamma\hbar^3 U[e_1, e_1, l_1] + \frac{1}{2}\beta\gamma^2\hbar^3 U[e_1, l_1, l_1] + \frac{1}{6}\gamma^3\hbar^3 U[l_1, l_1, l_1]$

$TD = 6; $\mathbb{O}\left[e^{\hbar\gamma l_1+\beta\hbar e_1}, \{l_1, e_1\}\to 1\right] == \mathbb{O}\left[e^{\hbar\gamma l_1+e^{\hbar\gamma}\beta\hbar e_1}, \{e_1, l_1\}\to 1\right]$

True

$TD = 6; $\mathbb{O}\left[e^{\hbar\gamma l_1+\beta\hbar f_1}, \{f_1, l_1\}\to 1\right] == \mathbb{O}\left[e^{\hbar\gamma l_1+e^{\hbar\gamma}\beta\hbar f_1}, \{l_1, f_1\}\to 1\right]$

True

$TD = 3; $\mathbb{O}\left[e^{\hbar(\beta e_1+\alpha f_1+\delta e_1 f_1)}, \{f_1, e_1\}\to 1\right]$

$\left(1 - \delta\hbar h_1 - \alpha\beta\hbar^2 h_1 + \delta^2\hbar^2 h_1^2 + 2\alpha\beta\delta\hbar^3 h_1^2 - \delta^3\hbar^3 h_1^3\right)U[\,] + \left(\beta\hbar - 2\beta\delta\hbar^2 h_1 - \alpha\beta^2\hbar^3 h_1 + 3\beta\delta^2\hbar^3 h_1^2\right)U[e_1] + \left(\alpha\hbar - 2\alpha\delta\hbar^2 h_1 - \alpha^2\beta\hbar^3 h_1 + 3\alpha\delta^2\hbar^3 h_1^2\right)U[f_1] + \left(\frac{\beta^2\hbar^2}{2} - \frac{3}{2}\beta^2\delta\hbar^3 h_1\right)U[e_1, e_1] + \left(\delta\hbar + \alpha\beta\hbar^2 - 2\delta^2\hbar^2 h_1 - 4\alpha\beta\delta\hbar^3 h_1 + 3\delta^3\hbar^3 h_1^2\right)U[e_1, f_1] + \left(\frac{\alpha^2\hbar^2}{2} - \frac{3}{2}\alpha^2\delta\hbar^3 h_1\right)U[f_1, f_1] + \frac{1}{6}\beta^3\hbar^3 U[e_1, e_1, e_1] + \left(\beta\delta\hbar^2 + \frac{1}{2}\alpha\beta^2\hbar^3 - 3\beta\delta^2\hbar^3 h_1\right)U[e_1, e_1, f_1] + \left(\alpha\delta\hbar^2 + \frac{1}{2}\alpha^2\beta\hbar^3 - 3\alpha\delta^2\hbar^3 h_1\right)U[e_1, f_1, f_1] + \frac{1}{6}\alpha^3\hbar^3 U[f_1, f_1, f_1] + \frac{1}{2}\beta^2\delta\hbar^3 U[e_1, e_1, e_1, f_1] + \left(\frac{\delta^2\hbar^2}{2} + \alpha\beta\delta\hbar^3 - \frac{3}{2}\delta^3\hbar^3 h_1\right)U[e_1, e_1, f_1, f_1] + \frac{1}{2}\alpha^2\delta\hbar^3 U[e_1, f_1, f_1, f_1] + \frac{1}{2}\beta\delta^2\hbar^3 U[e_1, e_1, e_1, f_1, f_1] + \frac{1}{2}\alpha\delta^2\hbar^3 U[e_1, e_1, f_1, f_1, f_1] + \frac{1}{6}\delta^3\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]$

$TD = 6; With$\left[\left\{\nu = (1 + \hbar h\delta)^{-1}\right\},\right.$
$\mathbb{O}\left[e^{\hbar(\beta e_1+\alpha f_1+\delta e_1 f_1)}, \{f_1, e_1\}\to 1\right] == \mathbb{O}\left[\nu\, e^{\hbar\nu(-\hbar h\alpha\beta+\beta e_1+\alpha f_1+\delta e_1 f_1)}, \{e_1, f_1\}\to 1\right]$ /. $h_1 \to h$
]

True